# NAG Library Routine Document

# G05LEF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

G05LEF generates a vector of pseudorandom numbers taken from a beta distribution with $a$ and $b$.

## 2 Specification

```
SUBROUTINE G05LEF(A, B, N, X, IGEN, ISEED, IFAIL)
INTEGER           N, IGEN, ISEED(4), IFAIL
double precision  A, B, X(N)
```

## 3 Description

The beta distribution has PDF (probability density function)

$$f(x) = \frac{\Gamma(a+b)}{\Gamma(a)\Gamma(b)} x^{a-1}(1-x)^{b-1}, \quad \text{if } 0 \leq x \leq 1; a, b > 0.0,$$

$$f(x) = 0 \qquad \text{otherwise.}$$

One of four algorithms is used to generate the variates depending on the values of $a$ and $b$. Let $\alpha$ be the maximum and $\beta$ be the minimum of $a$ and $b$. Then the algorithms are as follows:

(i) If $\alpha < 0.5$, Jhnk's algorithm is used, see for example Dagpunar (1988). This generates the beta variate as $u_1^{1/a} / \left( u_1^{1/a} + u_2^{1/b} \right)$, where $u_1$ and $u_2$ are uniformly distributed random variates;

(ii) If $\beta > 1$, the algorithm BB given by Cheng (1978) is used. This involves the generation of an observation from a beta distribution of the second kind by the envelope rejection method using a log-logistic target distribution and then transforming it to a beta variate;

(iii) If $\alpha > 1$ and $\beta < 1$, the switching algorithm given by Atkinson (1979) is used. The two target distributions used are $f_1(x) = \beta x^{\beta}$ and $f_2(x) = \alpha(1-x)^{\beta-1}$, along with the approximation to the switching parameter of $t = (1-\beta)/(\alpha+1-\beta)$;

(iv) In all other cases, Cheng's BC algorithm (see Cheng (1978)) is used with modifications suggested by Dagpunar (1988). This algorithm is similar to BB, used when $\beta > 1$, but is tuned for small values of $a$ and $b$.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05LEF.

## 4 References

Atkinson A C (1979) A family of switching algorithms for the computer generation of beta random variates *Biometrika* **66** 141–5

Cheng R C H (1978) Generating beta variates with nonintegral shape parameters *Comm. ACM* **21** 317–322

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

# 5 Parameters

1: A – ***double precision*** *Input*

*On entry*: $a$, the parameter of the beta distribution.

*Constraint*: A $> 0.0$.

2: B – ***double precision*** *Input*

*On entry*: $b$, the parameter of the beta distribution.

*Constraint*: B $> 0.0$.

3: N – INTEGER *Input*

*On entry*: $n$, the number of pseudorandom numbers to be generated.

*Constraint*: N $\geq 0$.

4: X(N) – ***double precision*** array *Output*

*On exit*: the $n$ pseudorandom numbers from the specified beta distribution.

5: IGEN – INTEGER *Input*

*On entry*: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.

6: ISEED(4) – INTEGER array *Communication Array*

*On entry*: contains values which define the current state of the selected generator.

*On exit*: contains updated values defining the new state of the selected generator.

7: IFAIL – INTEGER *Input/Output*

*On entry*: IFAIL must be set to $0$, $-1$ or $1$. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

*On exit*: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value $-1$ or $1$ is recommended. If the output of error messages is undesirable, then the value $1$ is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is $0$. **When the value $-1$ or $1$ is used it is essential to test the value of IFAIL on exit.**

# 6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or $-1$, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, A $\leq 0.0$.

IFAIL $= 2$

On entry, B $\leq 0.0$.

IFAIL $= 3$

On entry, N $< 0$.

## 7 Accuracy

Not applicable.

## 8 Further Comments

To generate an observation, $y$, from the beta distribution of the second kind from an observation, $x$, generated by G05LEF the transformation, $y = x/(1-x)$, may be used.

## 9 Example

This example prints a set of five pseudorandom numbers from a beta distribution with parameters $a = 2.0$ and $b = 2.0$, generated by a single call to G05LEF, after initialization by G05KBF.

### 9.1 Program Text

```
*     G05LEF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          N
      PARAMETER        (N=5)
*     .. Local Scalars ..
      DOUBLE PRECISION A, B
      INTEGER          IFAIL, IGEN, J
*     .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          ISEED(4)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05LEF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05LEF Example Program Results'
      WRITE (NOUT,*)
*     Initialize the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 42344
      ISEED(4) = 742355
*     IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*
      A = 2.0D0
      B = 2.0D0
      IFAIL = 1
*
      CALL G05LEF(A,B,N,X,IGEN,ISEED,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
*
         WRITE (NOUT,99999) A, B
         WRITE (NOUT,99998) (X(J),J=1,N)
*
      ELSE
         WRITE (NOUT,99997) IFAIL
      END IF
*
99999 FORMAT (1X,'Beta Dist --- A=',F3.1,', B=',F3.1)
99998 FORMAT (1X,F10.4)
99997 FORMAT (1X,' ** G05LEF returned with IFAIL = ',I5)
      END
```

### 9.2 Program Data

None.

## 9.3 Program Results

```
 G05LEF Example Program Results

 Beta Dist --- A=2.0, B=2.0
     0.4334
     0.8888
     0.5604
     0.3799
     0.5064
```